


PAR. Lab 2

Lunar Lockout Game

Game Pieces

The lunar lockout game has several pieces. First, we have a 5x5 game board with a red square marked in the middle. Next, we have 5 helper spacecraft in various colours and then 1 red spacecraft. We are also given cards that specify initial setup position for some subset of spacecraft. On the back of each card is a solution.



Game Description

The goal of this game is to move the red spacecraft to the centre red square. One can move any spacecraft but they are limited to moving up-down or left-right. Whenever a spacecraft moves, it continues moving until it hits another spacecraft. Note that it is not legal for the spacecraft to move on the board. Therefore, all spacecraft need to cooperate together to stay on the board and help the red spacecraft reach the goal.

Game Play

We present two simple beginner puzzles to show typical game play. Puzzle 1 is the beginner puzzle shown above. Starting from the initial setup, the red spacecraft moves up, left, down, then left to finally end up at the goal position. In this simple puzzle, only the red robot had to move. In the more complex puzzles, we also require other helper spacecraft to move. Puzzle 2 is an example requiring another helper robot to move as well. In this assignment, you will be defining a representation for this puzzle and also implementing a planner to solve them.





Figure 2: Puzzle 1. The solution only requires moving the red spacecraft.



(a) Init (b) P - Move Down (c) P - Move Right (d) R - Move Right (e) R - Move Down
Figure 3: Puzzle 2. The solution requires first moving the purple spacecraft and then the red Spacecraft.

Define a way to represent the lunar lockout world (objects, states, actions)?

Objects

In the **lunar lockout world** game, we have spacecrafts located in certain positions of the board, and a position can be always defined by a pair (x, y), where the x represents the columns and y represents the row. Therefore, we can represent the world using two types of objects: spacecrafts and coordinates, but since our board is square we can use one type of object called position. If we have a board of 5x5 and 5 spacecrafts the objects will be:

- Spacecrafts: red yellow orange purple green
- Positions: n1 n2 n3 n4 n5

States

Our state will be the position of a spacecraft on the board. This will be represented by a spacecraft and two positions: the first will be the x-coordinate and the second will be the y-coordinate.

Actions

The set of actions will be to move a spacecraft UP, DOWN, LEFT and RIGHT. The parameters of the actions will be the spacecraft to be moved, the spacecraft that will stop the moving spacecraft, and the positions of both. The preconditions to perform the actions will be:

- The position of the stop craft and the position of the moving craft are on the same column or row and the stop craft will “stop” the moving craft (depending on the type of action: moving up/down or right/left).
- There is no other craft in the middle of the moving craft and the stop craft. To check this precondition we can use different strategies defined in the Domain axioms.

Explain the number of possible states for the suggested representations that you came up with?

The final number of possible states follows the formula of the permutations without repetitions, therefore:

$$N_States = (positions\ in\ the\ board)! / (positions\ in\ the\ board - number\ of\ spacecrafts)! ,$$

In our representation, we have 5 spacecrafts and 25 positions in the board:

$$N_oStates = 25! / (25 - 5)! = 6375600$$

Write the PDDL files required for a planner of your choice to solve this problem (Hint: a state-space search is likely the most simple).

Many valid solutions to this problem. We were looking for clear idea about the predicates that a state could be composed of. We also wanted to see recognition that actions have precondition, add, and delete effects. Domain axioms should be stated but are implicit in the definition of actions. Below is a possible solution, represented using PDDL.

Domain PDDL file:

Listing 1: Domain specification

```
(define (domain lunar-lockout-b)
  (:requirements :strips :adl :typing)

  (:types spacecraft coord)

  (:predicates
    (greater-than ?s1 - coord ?s2 - coord)
    (one-greater-than ?s1 - coord ?s2 - coord)
    (at ?spacecraft - spacecraft ?row - coord ?col - coord))

  (:action move-left
    :parameters (?spacecraft1 - spacecraft ?row - coord
                  ?spacecraft2 - spacecraft ?s2-col - coord
                  ?from-col - coord ?to-col - coord)
    :precondition (and
      (at ?spacecraft1 ?row ?from-col)
      (at ?spacecraft2 ?row ?s2-col)
      (one-greater-than ?to-col ?s2-col)
      (greater-than ?from-col ?to-col)
      (forall (?spacecraft3 - spacecraft ?col - coord)
        (not (and
          (at ?spacecraft3 ?row ?col)
          (greater-than ?from-col ?col)
          (greater-than ?col ?s2-col)))))
    :effect (and (at ?spacecraft1 ?row ?to-col)
      (not (at ?spacecraft1 ?row ?from-col))))

  (:action move-right
    :parameters (?spacecraft1 - spacecraft ?row - coord
                  ?spacecraft2 - spacecraft ?s2-col - coord
                  ?from-col - coord ?to-col - coord)
    :precondition (and
      (at ?spacecraft1 ?row ?from-col)
      (at ?spacecraft2 ?row ?s2-col)
      (one-greater-than ?s2-col ?to-col)
      (greater-than ?to-col ?from-col)
      (forall (?spacecraft3 - spacecraft ?col - coord)
        (not (and
          (at ?spacecraft3 ?row ?col)
          (greater-than ?col ?from-col)
          (greater-than ?s2-col ?col)))))
    :effect (and (at ?spacecraft1 ?row ?to-col)
      (not (at ?spacecraft1 ?row ?from-col))))
```

```

(:action move-up
:parameters (?spacecraft1 - spacecraft ?col - coord
             ?spacecraft2 - spacecraft ?s2-row - coord
             ?from-row - coord ?to-row - coord)
:precondition (and
               (at ?spacecraft1 ?from-row ?col)
               (at ?spacecraft2 ?s2-row ?col)
               (one-greater-than ?s2-row ?to-row)
               (greater-than ?to-row ?from-row)
               (forall (?spacecraft3 - spacecraft ?row - coord)
                (not (and
                      (at ?spacecraft3 ?row ?col)
                      (greater-than ?row ?from-row)
                      (greater-than ?s2-row ?row))))))
:effect (and (at ?spacecraft1 ?to-row ?col)
             (not (at ?spacecraft1 ?from-row ?col))))

(:action move-down
:parameters (?spacecraft1 - spacecraft ?col - coord
             ?spacecraft2 - spacecraft ?s2-row - coord
             ?from-row - coord ?to-row - coord)
:precondition (and
               (at ?spacecraft1 ?from-row ?col)
               (at ?spacecraft2 ?s2-row ?col)
               (one-greater-than ?to-row ?s2-row)
               (greater-than ?from-row ?to-row)
               (forall (?spacecraft3 - spacecraft ?row - coord)
                (not (and
                      (at ?spacecraft3 ?row ?col)
                      (greater-than ?from-row ?row)
                      (greater-than ?row ?s2-row))))))
:effect (and (at ?spacecraft1 ?to-row ?col)
             (not (at ?spacecraft1 ?from-row ?col))))

```

Problem PDDL file:

Listing 2: Problem 1 specification

```
(define (problem lunar-lockout-b-probone)
  (:domain lunar-lockout-b)
  (:objects red yellow purple orange green - spacecraft
            zero one two three four - coord)
  (:goal (at red two two))
  (:init
    (at red zero four)
    (at yellow one three)
    (at orange four four)
    (at green three two)
    (at purple two one)

    (greater-than one zero)
    (greater-than two zero)
    (greater-than three zero)
    (greater-than four zero)

    (greater-than two one)
    (greater-than three one)
    (greater-than four one)

    (greater-than three two)
    (greater-than four two)

    (greater-than four three)

    (one-greater-than one zero)
    (one-greater-than two one)
    (one-greater-than three two)
    (one-greater-than four three)))
```

Exercises

1. Write a problem PDDL file for the puzzle 2 shown in figure 3?
2. Is there any difference in the size of the states of puzzle 1 and puzzle 2?
3. Define another way to represent the lunar lockout world (objects, states, actions).

Ensure there is a big difference in the set of possible states between your representation and the representation explained above.